

Cursillo avanzado

GeoGebra 3.1.226.0

Ficha de ANIMACIÓN MANUAL Y AUTOMÁTICA

NOTA: con MacOS, Clic derecho = Ctrl + Clic

Paso previo: Si es preciso para limpiar la pantalla de una construcción anterior, en el menú **Archivo** pulsar sobre **Nuevo** (en el cuadro de diálogo emergente escoger "No Guardar"). Observar que inicialmente se encuentra elegida la Herramienta fundamental: **Elige y Mueve**.

Acción	Detalles
Abrir archivo 1_A.ggb	
(Explicación)	 Aparece una circunferencia de centro A que pasa por B, un punto en ella C y un radio r. Tenemos dos deslizadores angulares, α y β. GeoGebra no permite, de momento, animar de forma desatendida los objetos, pero sí los deslizadores. Esta acción no es tan natural como en Cabri, pero tiene la gran ventaja de ofrecer un control total sobre la animación.
Arrastrar A, B, C, c y r.	Elige y Mueve. Clic en cada objeto y mover sin soltar. El segmento "r" no se dejará arrastrar. (Debemos acostumbrarnos a volver siempre a esta herramienta general después de usar las específicas de creación de objetos.)
Mover C con teclas flecha	Observar la velocidad variable según la posición de C en c (puede ser nula).
Mover C con teclas + y -	Observar ahora la velocidad constante del movimiento de C.
Mover C con teclas + y - manteniendo pulsadas mayús, ctrl y alt	Esas teclas multiplican la velocidad por un factor determinado. Tecla mayús : factor 1/10 Tecla ctrl : factor 10 Tecla alt : factor 100
A + (r; α)	Nota: El punto y coma indica coordenadas polares.
Cambiar estilo	Clic derecho sobre D, ya sea en Vista Algebraica o en Vista Gráfica. Propiedades. Pestaña Color. Elegir naranja.
D + (r/4; 2β+30°)	Aparecerá un punto E.
Cambiar estilo	Clic derecho sobre E. Propiedades. Pestaña Color. Elegir naranja.
Activar Rastro de E	Clic derecho sobre E. Activa Rastro. Nota: Ctrl+F borra los rastros dejados.
Animar el deslizador α	Clic derecho sobre α . Propiedades. Pestaña Deslizador. Elegir, en animación, Repite Incremento. Activar Animación Automática.
Animar el deslizador β	Clic derecho sobre β. (De forma predefinida, está elegido, en animación, Repite Oscilante.) Activar Animación Automática.
Parar / Reproducir	Botón situado en la esquina inferior derecha de la Vista Gráfica.
Desactivar la animación automática de α y β	Clic derecho sobre α . Desactivar Animación Automática. Clic derecho sobre β . Desactivar Animación Automática.

Construcción completa: 1.ggb

Ficha de SEÑALAR POSICIONES CON PRECISIÓN GRACIAS AL ZOOM

Acción	Detalles
Abrir archivo 2_A.ggb	
(Explicación)	GeoGebra nos ayuda a descubrir . Veamos un ejemplo. Se trata de maximizar el área de intersección de un círculo y un triángulo arbitrario, pero que aquí consideramos acutángulo para simplificar. (*)
	Primero hemos cuantificado el área de intersección descomponiéndola en triángulos y sectores, y sumando todo. Esto es demasiado extenso para verlo aquí con más detalle, pues aparecen distintas situaciones.
	Luego, hemos encontrado que esa área será máxima cuando el centro del círculo se encuentre en una trayectoria entre el Incentro y el Circuncentro del triángulo, pero que su posición exacta depende del radio del círculo.
	Usando el Zoom de Acercamiento, hemos dejado unos puntos blancos muy precisos indicando dónde se alcanza el máximo según el tamaño del radio (r).
	Falta establecer el lugar geométrico que corresponde a esa trayectoria.
(1ª Conjetura: trayectoria circular)	Al trazar una circunferencia que pase por el Incentro, el Circuncentro y un punto de la trayectoria (P59), los demás parecen ajustarse a ese arco circular.
Zoom de acercamiento sobre los puntos de la trayectoria	Sin embargo, al realizar un zoom de varios aumentos sobre los puntos blancos, vemos que no es así.
2 a = b Comprobar si los puntos yacen en la circunferencia	Además, si elegimos la herramienta Relación entre Dos Objetos (desplegar penúltimo botón) y hacemos clic sobre un punto blanco (distinto de P59) y sobre la circunferencia azul podemos comprobar que esos puntos no descansan en la circunferencia.
Abrir archivo 2_B.ggb	
(Explicación)	Una vez desestimada la trayectoria circular, intentamos aproximarla por otra curva. Tras varios ensayos, probamos con la hipérbola de Stammler , que pasa por el Incentro I, el Circuncentro O y los tres excentros E_A , E_B y E_C (centros de los círculos tangentes a las prolongaciones de los lados).
Cónica por I, O, E_A , E_B y E_C	Cónica dados Cinco de sus Puntos. Clic en esos cinco puntos.
Zoom de acercamiento sobre los puntos de la trayectoria	Ahora, al hacer zoom de varios aumentos, vemos que el ajuste es extremadamente preciso.
a = b Comprobar si los puntos yacen en la hipérbola	Además, si elegimos la herramienta Relación entre Dos Objetos y hacemos clic sobre un punto blanco y sobre la hipérbola podemos comprobar (no demostrar) que esos puntos yacen sobre ella.

(*) El enunciado completo de nuestra conjetura sería:

Dado un triángulo no obtusángulo, todos los círculos no triviales que maximizan el área de intersección con el triángulo tienen sus centros en el arco hiperbólico que une el incentro con el circuncentro sobre la hipérbola de Stammler.

Si el triángulo es obtusángulo, esos centros se sitúan en el segmento MH y en el arco hiperbólico HI sobre la hipérbola de Stammler, siendo M el punto medio del lado mayor del triángulo y **H** el punto de intersección de la hipérbola con el segmento que une M con el punto de Lemoine.

Nota: Entendemos aquí por "círculo no trivial" cualquiera cuyo radio esté comprendido entre el radio de la circunferencia inscrita y el radio de la circunferencia circunscrita al triángulo.

Es decir, dado un triángulo, entenderemos por "círculo trivial" aquel cuyo radio es menor que el de la circunferencia inscrita o mayor que el de la circunferencia circunscrita. En el primer caso, existen infinidad de puntos (uno de ellos el incentro) que tomados como centro del círculo lo mantienen en el interior del triángulo. En el segundo caso, existen infinidad de puntos (uno de ellos el circuncentro) que tomados como centro de ellos el circuncentro) que tomados como centro del cínculo triángulo. En el segundo caso, existen infinidad de puntos (uno de ellos el circuncentro) que tomados como centro del círculo mantienen al triángulo en su interior. En ambos casos, para cualquiera de esos centros el área de solapamiento es, trivialmente, máxima.

Para más información:

http://geometriadinamica.es/Investigaciones/Lugares-geometricos/El-cristo-de-la-farola.html

http://divulgamat.ehu.es/weborriak/RecursosInternet/GeoDinamica/CristoFarola/Farola.asp

Ficha de NAVEGACIÓN

N٥	Acción	Detalles
	Abrir archivo 3_A.ggb	
1	Punto PosiciónTexto	Nuevo Punto . Clic en una posición cercana a la esquina superior derecha de la Vista Gráfica. Inmediatamente, después de hacer clic, escribir el nombre (PosiciónTexto).
	Ocultar PosiciónTexto	Clic derecho sobre el punto PosiciónTexto. Destildar Muestra Objeto.
2	Castaño = (1,2)	
3	Roble = (6,1)	
4	Roca = (4,-4)	
5	ABC Insertar texto1	Desplegar el penúltimo botón, elegir Inserta Texto y hacer clic en la Vista Gráfica. Editar el siguiente texto (el símbolo de los grados, al ser texto, se introduce desde el teclado): Caminar desde la Roca hasta el Castaño, girar 60° a la izquierda,
		caminar una distancia igual y marcar.
	Posicionar texto1	Clic derecho sobre el texto1. Propiedades. En la pestaña Posición elegir el punto PosiciónTexto.
6	Segmento Roca- Castaño	Segmento. Clic sobre Roca y sobre Castaño.
7	Rotar -60º el punto Roca respecto a Castaño y nombrar como M1	Desplegar el antepenúltimo botón y elegir Rota Objeto en torno a un Punto, el Ángulo indicado. Clic en Roca, clic en Castaño, introducir 300 (evitando borrar el símbolo de grados). Inmediatamente, escribir el nombre (M1).
8	Ángulo M1- Castaño-Roca	Ángulo. Clic en M1, Castaño y Roca.
	Adornar ángulo	Clic derecho sobre el ángulo anterior. Propiedades. En la pestaña Básico, elegir Muestra Rótulo -> Valor. En la pestaña Decoración, el último adorno.
	Abrir archivo 3_B.ggb	
9	(Explicación)	Se ha creado el segmento M1-Castaño.
10	(Explicación)	Se ha insertado y posicionado un texto2.
11	(Explicación)	Se ha creado el texto Roca-Roble.

12	(Explicación)	Se ha rotado 120º en sentido Antihorario el punto Roca respecto a Roble y se ha nombrado como M2.
13	(Explicación)	Se ha creado el ángulo Roca-Roble-M2.
14	(Explicación)	Se ha creado el segmento M2-Roble.
15	(Explicación)	Se ha insertado y posicionado un texto3.
16	(Explicación)	Se ha creado el segmento M1-M2.
17	Punto Medio M1- M2, nombrar como TESORO	Elegir Punto Medio y hacer clic en el segmento anterior. Inmediatamente, escribir el nombre (TESORO).
	Ocultar nombres de los segmentos	Clic derecho sobre un segmento, Propiedades. En la ventana de la izquierda, pulsar sobre Segmento (esto seleccionará todos). En la pestaña Básico, destildar Muestra Rótulo.
	Mostrar la Barra de Navegación	En el menú Vista, activar Barra de Navegación por Pasos de Construcción.
	Abrir Protocolo de la Construcción	Pulsar el botón situado a la derecha de la Barra de Navegación. Ensanchar un poco la ventana para mejorar la lectura de los nombres.
	Mostrar Puntos de Interrupción	En la ventana del Protocolo, menú Vista, activar Punto de Interrupción. Tildar todas las casillas excepto las 4 primeras, la 7 y la 12.
	Imponer condición de visibilidad a texto1	En la ventana del Protocolo, Clic derecho sobre texto1, Propiedades, Pestaña Avanzado, escribir como condición para exponer el objeto:
		PasoConstrucción[]<10
		Pulsar Cierra.
	Abrir archivo 3_C.ggb	
	(Explicación)	Se ha insertado una imagen de fondo (al establecerse como fondo se visualiza en cualquier paso, no es necesario colocarla al principio) y se han impuesto las siguientes condiciones para exponer el objeto:
		Para el punto M1: PasoConstrucción[]>9 Para el punto M2: PasoConstrucción[]>14 Para el texto 2: PasoConstrucción[]<15
	Mostrar Sólo los Puntos de Interrupción	En la ventana del Protocolo, menú Vista, activar Mostrar Sólo los Puntos de Interrupción.
	Cerrar el Protocolo	Clic sobre el aspa de cierre de ventana.
	Usar la Barra de Navegación.	Pulsar el primer botón de la Barra para ir al primer paso y pulsar sucesivas veces el botón de avance.

Construcción completa: 03.ggb

Se ha elegido esta construcción porque plantea la siguiente cuestión:

No importa desde donde comencemos la búsqueda del Tesoro, es decir, la posición de la Roca es irrelevante. Si movemos la Roca de lugar, dejando en paz a los árboles, jel tesoro permanece en el mismo sitio! (Investiga por qué. En el archivo 3_D.ggb tienes una pista.)

Ficha de GEOMETRÍA INTERNA

Acción	Detalles
Abrir archivo 4_A.ggb	
(Explicación)	Una recta definida por un punto O y un vector v tiene por ecuación vectorial: X = O + λ v
	Al colocar un punto C en esa recta, GeoGebra le asigna un valor λ_{C} del parámetro $\lambda.$
	Si C coincide con O, entonces $\lambda_c = 0$ (comprobar moviendo C). Si C coincide con O + v, entonces $\lambda_c = 1$ (comprobar moviendo C). En cualquier caso, $\lambda_c = RazónSimple[O, O+v, C]$
	Si variamos el módulo de v, el parámetro λ_c se mantiene constante, por lo que varía la posición absoluta de C. Es decir, GeoGebra toma el módulo del vector director v como "vector base" de la recta.
Abrir archivo 4_B.ggb	
(Explicación)	Una recta definida por dos puntos A y B equivale a estar definida por el punto A y el vector v = AB. Ecuación vectorial: X = A + λ v
	Al colocar un punto C en esa recta, GeoGebra le asigna un valor λ_{C} del parámetro $\lambda.$
	Si C coincide con A, entonces $\lambda_c = 0$ Si C coincide con B, entonces $\lambda_c = 1$ En cualquier caso, $\lambda_c = RazónSimple[A, B, C]$
	Si variamos la posición de B, el parámetro λ_{C} se mantiene constante, por lo que varía la posición de C.
Abrir archivo 4_C.ggb	
(Explicación)	GeoGebra usa el vector director v de la recta para establecer el vector director de cualquier recta paralela (v) y el vector director de cualquier recta perpendicular (n, normal a v).
	Así que al variar el módulo de v (por ejemplo, desplazando B) todos los puntos de las tres rectas (r, s y t) se desplazan, ya que la posición de todos esos puntos depende de A y B.
Abrir archivo 4_D.ggb	
Bisectrices (a y b) de r y t	Elegir Bisectriz y hacer clic en la recta r y en la recta t.
Mostrar las rectas a y b en forma vectorial	Clic derecho sobre cada recta. Elegir Forma Paramétrica.
ur = VectorU y pulsar F1	Mostrar la ayuda a los comandos proporcionada por la tecla F1
ur = VectorUnitario[r]	Una vez escrito ur = VectorU, pulsar Intro, escribir "r", pulsar Intro.
ut = VectorUnitario[t]	
ua = VectorUnitario[ur+ut]	

Punto D en a	Nuevo Punto. Clic sobre la recta a.
Mover B	
(Explicación)	Observemos que ahora, al mover B, la posición de D en la bisectriz no varía.
	GeoGebra toma Vectores Unitarios para hallar el vector director de la bisectriz, lo que provoca su independencia del módulo de v = AB (aunque sí depende de su sentido).
Abrir archivo 4_E.ggb	
(Explicación)	La recta definida por A y B tiene como vector director v = AB. Su ecuación vectorial es: X = A + λ v
	Su reflexión respecto al eje X tiene como vector director A'B'. Su ecuación vectorial es: X = O + λ A'B'
	Al colocar un punto Z en esa recta, GeoGebra le asigna un valor λ_Z del parámetro $\lambda.$
	Si Z coincide con O, entonces $\lambda_Z = 0$ Si Z coincide con O + A'B', entonces $\lambda_Z = 1$ (posición actual) En cualquier caso, $\lambda_Z = RazónSimple[O, O + A'B', Z]$
(Discusión)	¿Qué pasará al mover B?
	¿Se mantendrá Z en su posición o, por el contrario, será λ_z quien mantenga su valor constante y haga moverse a Z detrás de B'?
Mover B	

Ficha de CURVAS PARAMÉTRICAS Y DESLIZADORES CON TOPES PARAMÉTRICOS

Acción	Detalles
Abrir archivo 5_A1.ggb	(Curvas paramétricas. Inicio de construcción.)
(Explicación)	Tenemos un punto libre P , rojo, situado en (4, 1).
	Tenemos dos parámetros, a y b , que harán de extremos del intervalo [a, b] donde definiremos la curva, y tenemos otro parámetro k que formará parte de la definición de la curva.
	Tenemos también un parámetro $\mathbf{h_0}$, que varía entre 0 y 1.
	Por último, hemos definido dos funciones u y v :
	$u(x) = e^{(-k x)} cos(x)$ $v(x) = e^{(-k x)} sin(x)$
	Nota: al definir estas funciones, se crea automáticamente el valor constante e = 2.71 como objeto auxiliar.
	La curva (u(t), v(t)) en cartesianas equivale a (e^(-k t); t) en polares.
Curva[x(P) + u(t), y(P) + v(t), t, a, b]	Se creará la curva (u(t), v(t)), desplazada hasta P.
h = a + (b - a) h_0 (*)	Este parámetro variará siempre en [a, b], pues h ₀ varía en [0,1].
(e^(-k h); h)	Se creará un punto A sobre la curva sin desplazar, en la posición t = h.
A + P	Se creará un punto B sobre la curva desplazada, en la posición t = h.
Abrir archivo 5_A2.ggb	(Curvas paramétricas. Construcción acabada.)
(Explicación)	Se ha activado el rastro del punto B y mejorado el estilo. Mover los deslizadores para observar su efecto.

(*) Además de h = a + (b - a) h₀, que varía siempre entre a y b, otras posibilidades son:

• h = Mínimo[a, b] + (Máximo[a, b] - Mínimo[a, b]) h₀

que varía entre a y b cuando a es menor o igual que b; si a es mayor que b, varía entre b y a.

• h = Mínimo[b, a + (b - a) h₀]

varía entre a y b cuando a es menor o igual que b. Si a es mayor que b, queda fijo en b.



Ficha de MOVIMIENTO BASADO EN PARÁMETROS

Acción	Detalles
Abrir archivo 5_B.ggb	(Movimiento basado en parámetros.)
(Explicación)	Usar la barra de Navegación.



Ficha de FUNCIONES A TROZOS

Acción	Detalles
Abrir archivo 5_C.ggb	(Funciones a trozos.)
(Explicación)	Podemos usar el comando Si para definir funciones a trozos. Estas funciones se comportan igual que las normales. En el comando Si, el uso de la variable x en la condición se comporta como un parámetro.
	$f(x) = Si[x<1, Función[f_1(x), 0, 1], Función[f_2(x), 1, 2]]$
Activar "Más trozos"	
(Explicación)	Si hay más de dos trozos, podemos anidar los comandos Si, o bien recurrir a funciones similares a las que Derive Ilama "CHI", que toman el valor 1 en cada intervalo de definición y 0 en el resto.
	$ \begin{array}{ll} t_1(x) = Si[x \ge 0 \&\& x < 1, 1, 0] & t_2(x) = Si[x \ge 1 \&\& x < 2, 1, 0] \\ t_3(x) = Si[x \ge 2 \&\& x < 3, 1, 0] & t_4(x) = Si[x \ge 3 \&\& x < 4, 1, 0] \end{array} $
	$g(x) = \text{Función}[t_1(x) f_1(x) + t_2(x) f_2(x) + t_3(x) f_3(x) + t_4(x) f_4(x), 0, 4]$



Ficha de COLOR PARAMÉTRICO

Acción	Detalles
Abrir archivo 5_D1.ggb	(Color paramétrico)
(Explicación)	Tenemos dos puntos A y B. Queremos conocer el lugar geométrico de los puntos C equidistantes de ellos (la mediatriz).
	Para ello, establecemos como Color Dinámico del punto C la expresión a / b para cada color del RGB y activamos el Rastro, donde a y b son las distancias respectivas a A y B.
	Cuando a / b sea 1 (o entero impar), el color resultante será blanco, lo que provocará que destaque, por contraste, sobre los demás.
Abrir archivo 5_D2.ggb	
(Explicación)	Ahora queremos conocer el lugar geométrico de los puntos C en los que el producto (p) de las distancias a A y B permanece constante.
	Para ello, establecemos como Color Dinámico del punto C la expresión p / k para cada color del RGB y activamos el Rastro.
	(El deslizador k nos permite ajustar el resultado a un número entre 0 y 1.)
Abrir archivo 5_D3.ggb	
(Explicación)	Variando ligeramente la expresión p/k, podemos conseguir lo mismo en color. Si conocemos la expresión paramétrica de la curva, y activamos el Rastro de ella, el resultado es la familia de los óvalos de Cassini.
	Rojo: p / k Verde: (p / k + 1) / 2 Azul: p / k - 1
Abrir archivo 5_D4.ggb	
(Explicación)	Las funciones periódicas pueden ayudarnos a conseguir variaciones de color cíclicas y contrastadas, permitiendo diferenciar mejor cada curva de la familia.
	Rojo: 0.5 + cos(p/k) sin(p/k) Verde: sin(p) Azul: cos(p/k)



Óvalos de Cassini: el producto de las distancias a dos focos es constante (la Lemniscata es un caso particular)







Ficha de 3D y Macros

Acción	Detalles
Abrir archivo 6_A.ggb	
(Explicación)	Tenemos dos ángulos independientes, α y β , que harán el papel de ángulo de inclinación y ángulo de rotación, respectivamente.
	Tenemos una lista (llamada puntoLista) con las tres coordenadas $\{p_x, p_y, p_z\}$ de un punto cualquiera (en este caso, $\{4,4,4\}$).
	Hemos construido la proyección al plano:
	$P444 = (p_x \sin(\beta) + p_y \cos(\beta),$ -p_x cos(\beta) sin(\alpha) + p_y sin(\beta) sin(\alpha) + p_z cos(\alpha))
	Después, hemos convertido esa proyección en una nueva herramienta, a través del menú Herramientas > Creación de Herramienta Nueva:
	Objeto de Salida: P444 Objetos de Entrada: α , β , puntoLista Nombre de Herramienta y de Comando: P3D
	Icono: / (imagen de 30x30 píxeles)
	Por último, en Herramientas > Gestión de Herramientas , hemos guardado la Macro como:
	mi_herramienta_P3D.ggt
	para poder abrirla desde cualquier otra construcción.
P400 = P3D[α, β, {4,0,0}]	
P040 = P3D[α , β , {0,4,0}]	
(etc.)	
Abrir archivo 6_B.ggb	
(Explicación)	De la misma forma, se han creado en total los ocho vértices de un cubo de arista 4 unidades:
	P000, P400, P040, P004, P440, P404, P044 y P444
	Sólo nos queda formar las caras con la herramienta Polígono.
Abrir archivo 6_C.ggb	
(Explicación)	Al mover los ángulos α y $\beta,$ los puntos proyectados simulan las tres dimensiones.

Ficha de TRES TRUCOS SENCILLOS

Acción	Detalles
Abrir archivo 7_A1.ggb	(Control del Rastro.)
Punto A	
B = A	
Ocultar rótulo de B y activar su rastro	
Crear Casilla de Control "Activa Rastro". Controlará la aparición de B.	Cubrir el cuadro de diálogo escribiendo "Activa Rastro" como subtítulo y eligiendo el punto B en la lista desplegable. Pulsar el botón Aplica.
Mover el punto A. Destildar la casilla de control. Volver a mover el punto A.	
Abrir archivo 7_A2.ggb	
(Explicación)	En esta construcción empleamos el truco anterior para crear algunos diseños.
Abrir archivo 7_B.ggb	(Elección excluyente.)
(Explicación)	Imponiendo las debidas condiciones de visibilidad a los propios controladores de visibilidad, podemos obtener casillas de control con opciones mutuamente excluyentes.
Activar casillas de control	
Abrir archivo 7_C.ggb	(Libertad controlada.)
(Explicación)	 Si el punto azul descansa sobre el segmento, ¿cómo es posible que dirija su movimiento? El truco reside en que el punto azul no descansa sobre el segmento, sino que es éste el que se construye a partir de aquél. Los extremos del deslizador a (0 y 1.5) provocan que el punto azul (r; α) tenga limitado su movimiento al disco de radio 1.5 centrado en el origen.
Mostrar Rótulo de B	
Mostrar los deslizadores r y α	

El hecho de poder activar y desactivar el rastro a voluntad, usando el primero de los trucos que hemos visto, unido al uso del color dinámico objeto, nos permite crear fácilmente diseños aparentemente diferentes, pero en realidad procedentes de una única construcción, como los siguientes.





Ficha de COMPLEJOS, LISTAS Y FRACTALES

Acción	Detalles
Abrir archivo 8_A.ggb	(Conjunto de Mandelbrot)
(Explicación)	Definimos la sucesión A, $A^2 + A$,, $A_{n+1} = A_n^2 + A$ (el punto A, afijo del complejo, se comporta operativamente como el propio número complejo; la potencia es la potencia compleja).
	El conjunto de Mandelbrot es el conjunto de los A para los cuales la sucesión de módulos de los complejos de la sucesión anterior está acotada.
	Colocamos un punto A y calculamos A ₁ ,, A ₂₀ :
	$A_1 = A$ $A_2 = A_1^2 + A$
	$A_{20} = A_{19}^{2} + A$
	y creamos la lista:
	complejos = {A ₁ , A ₂ , A ₃ ,, A ₁₉ , A ₂₀ }
	Hallamos el máximo de sus módulos:
	Máximo[EliminaIndefinidos[Secuencia[Longitud[Vector[Elemento[complejos, s]]],
	s, 1, Longitud[complejos]]]]
	Nota: La instrucción anterior, compuesta de comandos anidados, se interpreta así (se lee "de dentro hacia fuera"):
	 Toma el complejo que ocupa el lugar "s" de la lista "complejos" Obtén el vector de posición de ese afijo Halla su módulo (longitud) Crea la lista correspondiente a los pasos anteriores, cuando "s" varía entre 1 y la longitud de la lista "complejos". Si en la lista anterior hay indefinidos, elimínalos. De la lista final, halla el máximo.
	Asignamos un color entre 0 (negro) y 1 (blanco) a ese máximo:
	color = Si[max > 2, 1, max / 2]
	Damos al punto A el Color Dinámico RGB:
	[R] color [G] (1+color ²)/2 [B] (1+color)/2
Mover A	
Abrir archivo 8_B.ggb	
(Explicación)	Es el mismo archivo anterior, pero con zoom de aproximación en una región.
Mover A	



Conjunto de Mandelbrot con GeoGebra.



Conjunto de Mandelbrot con GeoGebra. Detalle del extremo superior.

Ficha de SUPERFICIES

Acción	Detalles
Abrir archivo 9.ggb	(Toro)
	En la ficha de 3D veíamos que podíamos proyectar un punto 3D $\{p_x, p_y, p_z\}$ como:
	$(p_x \sin(\beta) + p_y \cos(\beta), - p_x \cos(\beta) \sin(\alpha) + p_y \sin(\beta) \sin(\alpha) + p_z \cos(\alpha))$
	donde α y β eran los ángulos de inclinación y rotación.
	Si conocemos las ecuaciones paramétricas de una superficie en el espacio, (x(u,v), y(u,v), z(u,v)), podemos generar dos familias de curvas sin más que sustituir estas coordenadas en vez de (p_x , p_y , p_z) en la proyección anterior:
	list1 = Secuencia[Curva]
	$x(u,v) \sin(\beta) + y(u,v) \cos(\beta),$ - $x(u,v) \cos(\beta) \sin(\alpha) + y(u,v) \sin(\beta) \sin(\alpha) + z(u,v) \cos(\alpha),$
	v, v1, v2], u, u1, u2, paso]
	list2 = Secuencia[Curva[
	$x(u,v) sin(\beta) + y(u,v) cos(\beta),$ - $x(u,v) cos(\beta) sin(\alpha) + y(u,v) sin(\beta) sin(\alpha) + z(u,v) cos(\alpha),$
	u, u1, u2], v, v1, v2, paso]
	En la primera lista de curvas, cada curva hace variar únicamente el parámetro v en cada punto. La familia de curvas se obtiene al secuenciar el parámetro u. En la segunda lista, viceversa.
	Las constantes u1, u2, v1 y v2 son los valores mínimo y máximo para los que hacemos variar u y v.
	La constante "paso" se añade para hacer más o menos densa la red de curvas.
	En el caso del toro, la parametrización empleada es:
	$\begin{aligned} x(u,v) &= (R + r \cos(u)) \cos(v) \\ y(u,v) &= (R + r \cos(u)) \sin(v) \\ z(u,v) &= r \sin(u) \end{aligned}$
	En la carpeta "superficies" podemos encontrar muchas más.
Mover ligeramente los extremos de los segmentos de la parte superior de la pantalla.	Estos extremos modifican las constantes u1, u2, v1, v2 y "paso".

Ficha de HOJA DE CÁLCULO

Acción	Detalles
Abrir archivo 99 .ggb	
(Explicación)	Se ha insertado un texto con el contenido completo de un artículo (previamente, se ha eliminado cualquier formato del texto).
	Creamos la lista "códigos", consistente en todos los números ASCII correspondientes a las letras de ese texto, usando el comando CódigoDeTexto.
	En la columna A de la hoja de cálculo escribimos el alfabeto, con sus variantes ortográficas.
	En la columna B, usamos de nuevo el comando CódigoDeTexto para convertir en números la columna A.
	En la columna C, usamos el comando CuentaSi para contar cuántos códigos de la lista "códigos" coinciden con alguno de la misma fila en la columna B.
	Celda C28: Suma de C1 a C27
	En la columna D, calculamos el porcentaje "100 C / C28" de aparición de cada letra en el texto.
	Creamos la secuencia de los intervalos base {0, 1, 2, 3,, 27}:
	intervalos = Secuencia[n, n, 0, 27]
	Finalmente, creamos el histograma correspondiente:
	Histograma[intervalos, D1:D27]
Probar a editar el texto y sustituirlo por otro cualquiera.	

Sugerencias de actividades a partir de esta estadística:

1) Estabilidad de las frecuencias. Ley de los grandes números. Tomando diferentes textos (preferiblemente largos) en castellano, observaremos como, en esencia, la distribución se comporta como una "huella dactilar" del español, sin variaciones significativas que dependan del texto elegido (salvo textos excesivamente técnicos o jergas similares). Ver, como ejemplo, el archivo "población.ggb" de la carpeta "cuenta_letras", que compara la muestra anterior con estudios más amplios.

2) Descifrado. Se puede usar esa escasa variabilidad de las frecuencias relativas como ayuda para descifrar mensajes secretos basados en códigos de sustitución (cada letra, un símbolo), al estilo "El escarabajo de oro" de Poe.

3) Lingüística comparada. Variando el idioma de los textos (y añadiendo, si hace falta, en la primera columna más acentos gráficos en vocales u otras letras), podemos comparar la "huella dactilar" de cada uno. Contrastando las diferencias de las gráficas podemos deducir un mapa de cercanías, es decir, decidir cuáles están más cercanas y cuáles más alejadas, o dicho de otra forma, a partir de la "distancia gráfica", crear una topología grafo-idiomática. Se pueden ver, como ejemplo, los archivos de la carpeta "cuenta_letras", donde se comparan entre sí los idiomas italiano, español, inglés y alemán.

POSIBILIDADES DE USO DE GEOGEBRA (lista provisional: foro GD, xiv jaem)

Nota: lista recopilada, tras diversas intervenciones de varias personas, por Josep Lluís Cañadilla

1. Usuario:

a. Profesorado. b. Alumnado.

2. El programa y el entorno:

a. El propio programa

i. Permite crear herramientas personales (macros) que facilitan la creación de construcciones complejas.

ii. Permite analizar el modus operandi de las construcciones realizadas por otros colegas y alumnos, es decir, como medio de adquisición de nuevos recursos para plantear y resolver situaciones.

b. Como applet en una página web

i. Permite limitar las herramientas que los alumnos pueden usar (al estilo "calculadora con sólo algunas teclas").

ii. Permite crear actividades interactivas, aleatorias, autocorrectivas, puntuables, con registro de la actividad y el resultado.

3. Punto de partida:

a. En blanco. b. Una construcción a medias.

c. Una construcción completada.

4. Infraestructura:

a. Ordenador + cañón de proyección. b. Aula con ordenadores.

5. Uso:

a. Por parte del profesorado

i. Visualizar figuras que sirvan de estímulo y faciliten la comprensión de conceptos e ideas matemáticas.

ii. Poder manipular e introducir variaciones significativas en las figuras mostradas.

iii. Como apoyo a la resolución de problemas (en su planteamiento, resolución y revisión).

iv. Para crear rápidamente ilustraciones estáticas (en papel, por ejemplo) altamente ajustadas a los deseos personales o específicos (material como base de investigación, material de exámenes, etc.).

v. Para alcanzar fácilmente conjeturas comprobadas (aunque no demostradas) sobre situaciones muy difíciles de conjeturar por otros medios, es decir, como medio de investigación.

vi. Para satisfacción propia.

vii. Para aprender geometría.

b. Por parte del alumno en actividades propuestas por el profesor.

i. Tipos:

- 1. Actividades guiadas (en papel).
- 2. Actividades dirigidas (por el profesor).
- 3. Actividades abiertas.
- 4. Actividades de reproducción.
- 5. Actividades interactivas.
- ii. Objetivos:
 - 1. Realizar una construcción geométrica.
 - 2. Manipular, observar, describir, conjeturar, comprobar, ...
 - 3. Para la "búsqueda la solución".
 - 4. Utilización en la resolución de problemas.

5. Investigación geométrica: búsqueda de propiedades y características de objetos geométricos, diferentes formas de construir la misma figura...

6. Análisis geométrico de figuras, imágenes y construcciones ya hechas.

7. Hacer cálculos métricos (medida de distancias, longitudes, ángulos y áreas).

8. Crítica y desarrollo estético (creación o modificación de simetrías, patrones, etc.).

9. Interconectar distintos campos matemáticos (en particular, representaciones gráficas, numéricas y algebraicas).

10. Determinar las posibilidades de movimiento de una construcción (análisis de los grados de libertad, articulaciones, robótica, mecanismos, etc.).

11. Alcanzar "un pensamiento más elevado", ya que la ayuda que proporciona el ordenador puede relegar la parte mecánica de la construcción a un segundo plano (de la misma forma que el uso de la calculadora facilita centrarse en el corazón del problema en vez de en las operaciones algorítmicas).

INFORMACIÓN

- Curso virtual (en construcción, URL temporal): http://geometriadinamica.es/igc/
- Instituto GeoGebra de Cantabria (URL en proyecto): www.geogebra.es
- GeoGebra Pre-Release: <u>http://www.geogebra.org/cms/index.php?option=com_content&task=blogcatego</u> ry&id=74&Itemid=59
- Intergeo: http://i2geo.net/xwiki/bin/view/Main/
- G4D: http://geometriadinamica.es/
- Espacio de debate GD en las XIV JAEM: http://www.xivjaem.org/forum/viewforum.php?f=11



Rafael Losada Liste GeoGebra Institute Trainer Instituto GeoGebra de Cantabria

Contacto: <u>rafaelll@educastur.princast.es</u> Web: http://www.iespravia.com/rafa/rafa.htm



Centro Internacional de Encuentros Matemáticos Castro Urdiales, marzo 2009